
Tutorial Objectives

High-level abstractions for message-passing concurrent
programming

Session types as robust and expressive organizing force

Substructural logics as a foundation for concurrency

Concrete instantiation of ideas in one retro language,
Concurrent C0

Entry to literature

Solved problems and current questions

Tutorial Approach

Organized around specification and programming

Three examples

Message streams (prime number sieve)
Concurrent data structure (queue)
Shared service (message buffer)

Arrive at working code

Extract essence and relate to logic

Tutorial Outline

Part I: Programming in Concurrent C0

Message streams (prime number sieve)
Concurrent data structure (queue)

Part II: Substructural Logics

Linear sequent calculus
Correspondence with message-passing concurrency

Part III: Sharing

Stratified session types
Manifest sharing via adjunctions

Prime Number Sieve

A process count produces the stream of numbers
2, 3, 4, 5, . . . up to some limit

A process primes receives the first number p and passes it
on, since it must be prime

Then primes spawns a new filter process which removes
all multiples of p from its input stream and recurses

In steady state we have

one producer process (count)
one filter process for each prime number already output
(filter pi )
one process (primes) that outputs only primes

A Session Type for Streams

A data structure of lists might be described as

list = {cons : int× list, nil : 1}
cons(2, cons(3, . . . , nil())) : list

We describe a stream of integer messages along some
communication channel analogously

stream = ⊕{next : 〈!int ; stream〉, empty : 〈 〉}
next, 2, next, 3, . . . , empty

⊕{`1 : A1, . . . , `n : An} sends one of the `i and
continues according to Ai

〈A1 ; . . . ; An〉 describes a sequence of interactions
!int sends an integer
〈 〉 closes the channels

Creating a Stream (live: primes.c1)

choice stream {

<!int ; !choice stream> Next;

< > Empty;

};

typedef <!choice stream> stream;

stream $c count(int n) {

for (int i = 2; i < n; i++)

//invariant $c : stream

{

$c.Next; /* $c : <!int ; stream> */

send($c, i); /* $c : stream */

}

$c.Empty; /* $c : < > */

close($c);

}

Takeaways

!<tp> sends a value v : <tp>

!choice <name> sends a label (internal choice)

$<ch> represents channel variables

stream $l count(...) {...} forks a new process and
provides a fresh channel $l : stream each time it is
called

Session type of $l changes during communication

Channel types must be loop invariant

Closing a channel terminates the providing process

Using a Stream (live: primes.c1)

void print_stream(stream $s) {

while (true) {

switch ($s) {

case Empty: { /* $s : < > */

wait($s);

print("\n");

return;

}

case Next: { /* $s : <!int ; stream> */

int x = recv($s); /* $s : stream */

printint(x); print(" ");

break;

}

}

}

}

int main() {

stream $nats = count(100);

print_stream($nats);

return 0;

}

Takeaways

Client performs complementary actions to provider

switch ($<ch>) {...} receives and branches on label

<tp> x = recv($<ch>); receives a basic data value

Channels behave linearly:

Guarantees session fidelity
All messages must be consumed

Filtering a Stream (live: primes.c1)

stream $t filter(int p, stream $s) {

switch ($s) {

case Empty: {

wait($s);

$t.Empty; close($t);

}

case Next: {

int x = recv($s);

if (x % p != 0) {

$t.Next;

send($t, x);

}

$t = filter(p, $s); /* tail-call */

}

}

}

Takeaways

Processes always provide channels

Process may also use channels

Provider/client send/receive actions are complementary

Used channels must close before provided channels

Tail calls can be used instead of loops

Generating Primes (live: primes.c1)

stream $p primes(stream $s) {

switch ($s) {

case Empty: {

wait($s); $p.Empty; close($p);

}

case Next: {

int x = recv($s);

$p.Next; send($p, x);

stream $t = filter(x, $s);

$p = primes($t);

}

}

}

int main() {

stream $nats = count(100);

stream $primes = primes($nats);

print_stream($primes);

return 0;

}

Takeaways

$<ch> = <proc>(...); (spawn) creates fresh channel
provided by new process instance

$<ch1> = $<ch2> (forwarding)

Identifies channels $<ch1> and $<ch2>

Terminates provider of $<ch1>
Converse of spawn

Strong identification between a process and the channel it
provides

Prime sieve creates n + 2 (lightweight) processes to
produce the nth prime

Implementation uses threads (C) or goroutines (Go)

Tutorial Outline

Part I: Programming in Concurrent C0

Message streams (prime number sieve)
Concurrent data structure (queue)

Part II: Substructural Logics

Linear sequent calculus
Correspondence with message-passing concurrency

Part III: Sharing

Stratified session types
Manifest sharing via adjunctions

A Simple Buffer

So far, all messages flow in the same direction through
the network of processes

In contrast, a simple buffer process is responsive

receive Ins, 1, Ins, 7,Del,
send Some, 1
receive Ins, 8,Del,
send Some, 7,
receive Del,
send Some, 8,
receive Del
send None, (close)

Labels received signify an external choice

External Choice

External choice N{`1 : A1, . . . , `n : An} receives one of
the `i and continues according to Ai

?int receives an integer

The buffer interface:

buffer = N{Ins : 〈?int ; buffer〉,Del : buffer_response}
buffer_response = ⊕{Some : 〈!int ; buffer〉,None : 〈 〉}
Internal to the process, use a sequential imperative queue

Buffer Session Type (live: lbuffer.c1)

choice buffer {

<?int ; ?choice buffer> Ins;

<!choice buffer_response> Del;

};

choice buffer_response {

<!int ; ?choice buffer> Some;

< > None;

};

Sequential Queue Interface (live: queue.h0)

typedef struct queue* queue_t;

queue_t new_queue(int capacity)

//@requires 1 <= capacity && capacity < (1<<20);

//@ensures \result != NULL;

;

bool is_empty(queue_t q)

//@requires q != NULL;

;

bool is_full(queue_t q)

//@requires q != NULL;

;

/* enqueing will drop x if q full */

void enq(queue_t q, int x)

//@requires q != NULL;

;

/* dequeing will return 0 if q empty */

int deq(queue_t q)

//@requires q != NULL;

;

Buffer Implementation (live: lbuffer.c1)

buffer $b new_buffer(int capacity) {

queue_t q = new_queue(capacity);

while (true) {

switch ($b) {

case Ins: { /* $b : <?int ; buffer> */

int x = recv($b); /* $b : buffer */

enq(q,x);

break;

}

case Del: { /* $b : !choice buffer_response */

if (is_empty(q)) {

$b.None; close($b);

} else {

int x = deq(q);

$b.Some; send($b, x);

}

break;

}

}

}

}

Takeaways

Local process state may be complex

Responsive systems rely on interaction between external
and internal choice

Processes offering an external choice have a concurrent
object-oriented flavor

Buffer Client (live: lbuffer.c1)

int main () {

buffer $b = new_buffer(10);

$b.Ins; send($b,1);

// $b.Ins; send($b,7);

$b.Del;

switch ($b) {

case None: error("bad!");

case Some: {

assert(1 == recv($b));

break;

}

}

$b.Del;

switch ($b) {

case None: {

wait($b);

break;

}

case Some: error("very bad!");

}

print("Yes!\n");

return 0;

}

Tutorial Outline

Part I: Programming in Concurrent C0

Message streams (prime number sieve)
Concurrent data structure (queue)

Part II: Substructural Logics

Linear sequent calculus
Correspondence with message-passing concurrency

Part III: Sharing

Stratified session types
Manifest sharing via adjunctions

What Does This Have To Do
With Substructural Logic?

Linear Sequent Calculus

Linear sequent: from antecedents ∆ prove succedent C

A1, . . . ,An︸ ︷︷ ︸
∆

` C

Substructural: each antecedent must be used exactly
once in proof (no weakening or contraction)

Judgmental Rules

Identity: From antecedent A we can prove succedent A

A ` A
idA

Cut: If we can prove succedent A we are allowed to
assume antecedent A

∆ ` A ∆′,A ` C

∆′,∆ ` C
cutA

Harmony: identity∗ and cut are admissible

A Process Interpretation of Proofs

Each antecedent and the succedent represent a channel
for communication

x1 : A1, . . . , xn : An︸ ︷︷ ︸
∆

` P :: (z : C )

Process P represents the proof of ∆ ` C

Process P provides channel z : C

Process P uses channels xi : Ai

Cut as Spawn

Annotate rule with process expressions

∆ ` P :: (x : A) ∆′, x : A ` Q :: (z : C )

∆′,∆ ` (x = P ; Q) :: (z : C )
cut

Spawned process P provides along fresh channel x

Continuation Q is client of P , using x

Other available channels (in ∆′,∆) are distributed
between P and Q.
Example (from prime sieve):

stream $nats = count(100);

stream $primes = primes($nats);

Identity as Forward

Annotate rule with process expressions

y : A ` (x = y) :: (x : A)
id

Forwarding process (x = y) identifies x and y
Example (stream constructor):
stream $l cons(int x, stream $k) {

$l.Next; /* $k : stream |- $l : <!int ; stream> */

send($l, x); /* $k : stream |- $l : stream */

$l = $k

}

Aside: π-Calculus

Spawn x = P ; Q corresponds to parallel composition
with a private channel

(νx)(P | Q)

But the π-calculus does not express threads of control

Identification x = y does not have a direct analogue

Internal Choice

As right and left rules of the sequent calculus

∆ ` A

∆ ` A⊕ B
∨R1

∆ ` B

∆ ` A⊕ B
∨R2

∆′,A ` C ∆′,B ` C

∆′,A⊕ B ` C
∨L

Cut Reduction

Key step in showing harmony is cut reduction

Replaces a cut at a compound proposition by cut(s) at
smaller propositions

For example:

D
∆ ` A

∆ ` A ∨ B
∨R1

E1
∆′,A ` C

E2
∆′,B ` C

∆′,A ∨ B ` C
∨L

∆′,∆ ` C
cutA∨B

−→

D
∆ ` A

E1
∆′,A ` C

∆′,∆ ` C
cutA

Cut Reduction as the Engine of Computation

Cut reduction is sequent calculus counterpart of
substitution

Cut reduction is more fine-grained than substitution

Cut reduction is communication
One premise of the cut has information to impart to the
other premise

D
∆ ` A

∆ ` A ∨ B
∨R1

E1
∆′,A ` C

E2
∆′,B ` C

∆′,A ∨ B ` C
∨L

∆′,∆ ` C
cutA∨B

−→

D
∆ ` A

E1
∆′,A ` C

∆′,∆ ` C
cutA

Internal Choice as Sending a Label

As right and left rules of the sequent calculus

∆ ` P :: (x : A)

∆ ` (x .π1 ; P) :: (x : A⊕ B)
∨R1

∆ ` P :: (x : B)

∆ ` (x .π2 ; P) :: (x : A⊕ B)
∨R2

∆′, x : A ` Q1 :: (z : C ) ∆′, x : B ` Q2 :: (z : C )

∆′, x : A⊕ B ` case x (π1 ⇒ Q1 | π2 ⇒ Q2) :: (z : C )
∨L

Observe how the type of the channel x changes

Cut reduction as communication

(x .π1 ; P) | (case x (π1 ⇒ Q1 | π2 ⇒ Q2)) −→ P | Q1

(x .π2 ; P) | (case x (π1 ⇒ Q1 | π2 ⇒ Q2)) −→ P | Q2

Concrete syntax in CC0 uses switch

Generalize to Labeled Internal Choice

A⊕ B , ⊕{π1 : A, π2 : B}
Generalized left and right rules

(k ∈ L) ∆ ` P :: (x : Ak)

∆ ` (x .k ; P) :: (x : ⊕{` : A`}`∈L)
∨Rk

(∀` ∈ L) ∆′, x : A` ` Q` :: (z : C )

∆′, x : ⊕{` : A`}`∈L ` case x (`⇒ Q`)`∈L :: (z : C )
∨L

Generalized cut reduction

(x .k ; P) | (case x (`⇒ Q`)`∈L) −→ P | Qk

External Choice

Switches role of succedent (provider) and antecedent
(client)

As right and left rules of the sequent calculus

∆ ` A ∆ ` B

∆ ` A N B
NR

∆,A ` C

∆,A N B ` C
NL1

∆,B ` C

∆,A N B ` C
NL2

This time, the left rule has the information

External Choice as Receiving a Label

Generalize to labeled external choice

A N B , N{π1 : A, π2 : B}
Generalized left and right rules

(∀` ∈ L) ∆ ` P` :: (x : A`)

∆ ` case x (`⇒ P`)`∈L :: (x : N{` : A`}`∈L)
NR

(k ∈ L) ∆, x : Ak ` Q :: (z : C )

∆, x : N{` : A`}`∈L ` (x .k ; Q) :: (z : C )
NLk

Same reduction!

(case x (`⇒ P`)`∈L) | (x .k ; Q) −→ Pk | Q

Sending from client to provider

Multiplicative Unit

In sequent calculus

· ` 1
1R

∆′ ` C

∆′, 1 ` C
1L

Cut reduction

· ` 1
1R

E
∆′ ` C

∆′, 1 ` C
1L

∆′ ` C
cut1

−→
E

∆′ ` C

Unit as End of Session

Process assignment to proofs

· ` close(x) :: (x : 1)
1R

∆′ ` Q :: (z : C )

∆′, x : 1 ` (wait(x) ; Q) :: (z : C )
1L

Cut reduction to close channel and terminate process

close(x) | (wait(x) ; Q) −→ Q

Existential Quantification

In sequent calculus, for data types τ

v : τ ∆ ` A(v)

∆ ` ∃n:τ.A(n)
∃R

∆′,A(c) ` C

∆′, ∃n:τ.A(n) ` C
∃Lc

The ∃R rule has information and sends

v : τ ∆ ` P :: (x : A(v))

∆ ` (send(x , v) ; P) :: (x : ∃n:τ.A(n))
∃R

∆′, x : A(c) ` Q :: (z : C )

∆′, x : ∃n:τ.A(n) ` (c = recv(x) ; Q) :: (z : C )
∃Lc

Straightforward reduction

(send(x , v) ; P) | (c = recv(x) ; Q) −→ P | [v/c]Q

40 / 67



Universal Quantification

Dual to existential quantification

Provider will receive a basic value

Client will send a basic value

In CC0, neither ∃x :τ.A nor ∀x :τ.A supports type
dependence, that is, occurrence of x in A

Summary of Correspondence

Curry-Howard Isomorphism

Linear Propositions Session Types
Sequent Proofs Process Expressions
Cut Reduction Computation

Cut is spawn (parallel composition)

Identity is forward (channel identification)

Logical connectives, from the provider point of view

Proposition Session Type Action Cont
A⊕ B ⊕{` : A`}`∈L send a label k ∈ L Ak

A N B N{` : A`}`∈L branch on received k ∈ L Ak

1 〈 〉 end session –
∃x :τ.A 〈!τ ; A〉 send a value v : τ A
∀x :τ.A 〈?τ ; A〉 receive a value v : τ A

Delegation: Sending Channels along Channels

Extend Curry-Howard interpretation of multiplicative
linear connectives A⊗ B and A( B

Proposition Session Type Action Cont
A⊗ B 〈!A ; B〉 send a channel y : A B
A( B 〈?A ; B〉 receive a channel y : A B

A⊕ B ⊕{` : A`}`∈L send a label k ∈ L Ak

A N B N{` : A`}`∈L branch on received k ∈ L Ak

1 〈 〉 end session –
∃x :τ.A 〈!τ ; A〉 send a value v : τ A
∀x :τ.A 〈?τ ; A〉 receive a value v : τ A

Metatheoretic Properties

Theorem: (session fidelity / type preservation) All processes
in a configuration remain well-typed and agree on
the types of the channels connecting them.

Theorem: (deadlock freedom / global progress) If all linear
processes are blocked then the computation is
complete.

Conjecture: (local progress) [ongoing work] If all recursive
types are inductive or coinductive

(i) communication along channels of inductive
type will terminate, and

(ii) communication along channels of coinductive
type will be productive

Tutorial Outline

Part I: Programming in Concurrent C0

Message streams (prime number sieve)
Concurrent data structure (queue)

Part II: Substructural Logics

Linear sequent calculus
Correspondence with message-passing concurrency

Part III: Sharing

Stratified session types
Manifest sharing via adjunctions

Sharing

Missing so far, logically: !A

Missing so far, operationally: sharing

Could we have a shared buffer with multiple producers
and consumers?

So far all channels are linear: one provider, one client

Examples abound: key/value store, database, output
device, input device, . . .

Stratification

Stratify session types into linear and shared

Shared S ::= ↑A
ongoing research︷ ︸︸ ︷

| S1 → S2 | S1 × S2 | . . .
Linear A ::= ⊕{` : A`}`∈L | N{` : A`}`∈L

| 〈!τ ; A〉 | 〈?τ ; A〉
| 〈!A ; B〉 | 〈?A ; B〉
| ↓S

Distinguish linear and shared channels

Modeled on LNL [Benton’94]

Traditional linear logic !A = ↓↑A

Shared Buffer Interface

Sharing is manifest in the type!

The linear buffer interface:

buffer = N{Ins : 〈?int ; buffer〉,Del : buffer_response}
buffer_response = ⊕{Some : 〈!int ; buffer〉,None : 〈 〉}
The shared buffer interface:

sbuffer = ↑N{Ins : 〈?int ; ↓sbuffer〉,Del : buffer_response}
buffer_response = ⊕{Some : 〈!int ; ↓sbuffer〉,None : ↓sbuffer}

Operational Interpretation of Shifts (Provider)

Process and channels go through shared and linear phases

xS : ↑A, from the provider perspective

Multiple clients along shared channel xS
Accept request to be acquired by one client along xS
Interact exclusively according to linear session xL : A

xL : ↓S , from provider perspective

Detach from single client
Provide along resulting shared channel xS : S

The linear protocol between X = ↑...↓X models a critical
region with exclusive access to a shared resource

Operational Interpretation of Shifts (Client)

Client performs matching interactions

xS : ↑A, from client perspective

Acquire exclusive access along xS
Interact exclusively according to linear session xL : A

xL : ↓S , from client perspective

Release provider
Revert to becoming one of many clients of xS : S

50 / 67



Shared Buffer Interface (live: sbuffer.c1)

choice buffer {

<?int ; # ; ?choice buffer> Ins;

<!choice buffer_response> Del;

};

choice buffer_response {

<!int ; #; ?choice buffer> Some;

<# ; ?choice buffer> None;

};

typedef <?choice buffer> lbuffer;

typedef <# ; ?choice buffer> sbuffer;

Takeaways

In concrete syntax, we only articulate ↑A as <# ; A>

↓S is implicit

Shared Buffer Implementation (live: sbuffer.c1)

sbuffer #b new_buffer(int capacity) {

queue_t q = new_queue(capacity);

while (true) {

lbuffer $b = (lbuffer)#b; /* accept */

switch ($b) {

case Ins: { /* $b : <?int ; buffer> */

int x = recv($b); /* $b : buffer */

enq(q,x);

#b = (sbuffer)$b; /* detach */

break;

}

case Del: { /* $b : !choice buffer_response */

if (is_empty(q)) {

$b.None;

#b = (sbuffer)$b; /* detach */

} else {

int x = deq(q);

$b.Some; send($b, x); /* detach */

#b = (sbuffer)$b;

}

break;

}

}

}

}

Takeaways

Shared channels have form #<ch>

Accept is implemented as a cast $<ch> = (<tp>)#<ch>;

Detach is implemented as a cast #<ch> = (<tp>)$<ch>;

Shared Buffer Clients (file: sbuffer-test.c1)

/* producer, from init to limit by step */

<> $c producer(int init, int step, int limit, sbuffer #b) {

for (int i = init; i < limit; i = i+step)

//invariant #b : sbuffer

{

lbuffer $b = (lbuffer)#b; /* acquire */

$b.Ins; send($b, i);

#b = (sbuffer)$b; /* release */

}

close($c);

}

/* consumer, of n messages */

<> $c consumer(int n, sbuffer #b) {

while (n > 0)

//invariant #b : sbuffer

{

lbuffer $b = (lbuffer)#b;

$b.Del;

switch ($b) {

case None: {

print("."); flush();

#b = (sbuffer)$b;

break;

}

case Some: {

int x = recv($b);

print("<"); printint(x); flush();

n = n-1;

#b = (sbuffer)$b;

break;

}}}

print("\n"); close($c);

Testing a Shared Buffer (file: sbuffer-test.c1)

int main() {

sbuffer #b = new_buffer(1000);

<> $p1 = producer(0, 3, 30, #b);

/* next line to sequentialize producers/consumers */

// wait($p1);

<> $p2 = producer(1, 3, 30, #b);

// wait($p2);

<> $p3 = producer(2, 3, 30, #b);

// wait($p3);

<> $c = consumer(30, #b);

// wait($c);

wait($p1);

wait($p2);

wait($p3);

wait($c);

return 0;

}

Takeaways

Shared buffers are not treated linearly

For session fidelity (type safety), type must be
equisynchronizing

If released, must be at the same type at which it was
acquired
Otherwise, waiting clients and provider may disagree on
the shared channels type
Could relax the restriction, with runtime type checking

Logical Interpretation

↑ and ↓ form an adjunction [Benton’94]

↓↑A is a comonad (!A)

↑↓S is a strong monad (©A)

Generalized in adjoint logic [Reed’09][Chargin et al.’17]

Adjoint propositions as stratified session types
Adjoint proofs as concurrent program
But: computation is not just proof reduction

Proof Construction and Deconstruction

Matching accept/acquire is seen as constructing a proof
by cut

This proof will be reduced with cut reduction until . . .

Matching detach/release is seen as deconstructing a cut
into two separate proofs

Shared channels limit nondeterminism in proof
construction

Shared processes are garbage-collected (reference
counting clients)

Deadlock is now possible!

Metatheoretic Properties, Including Sharing

Theorem: (session fidelity / type preservation) All processes
in a configuration remain well-typed and agree on
the types of the channels connecting them.

Theorem: (characterizing deadlocks / “progress”) If all
linear processes are blocked then

(i) either computation is complete, or
(ii) all linear processes are waiting for a response

to an acquire request (deadlock)

Dining Philosophers (files: dining_philosophers*.c1)

Summary: Linear Logic and Message-Passing

Curry-Howard interpretation of intuitionistic linear logic
[Caires & Pf’10]

Cut as parallel composition with private channel (spawn)
Identity as channel identification (forward)
Linear propositions as session types
Sequent proofs as process expressions
Cut reduction as communication
Guarantees session fidelity (preservation), local progress,
and termination

Extend to recursive types and processes [Toninho et al.’13]

Guarantee session fidelity and deadlock freedom (global
progress)
Inductive and coinductive types [ongoing work]

Summary: Linear Logic and Message-Passing

Extend further to permit sharing [Balzer & Pf’17]

Many more practical programs
Interleave proof construction, reduction, deconstruction
Proof construction may fail (deadlock)

Summary: Concurrent C0

C0: type-safe and memory-safe subset of C

Extended with a layer of contracts
Using in first-year imperative programming course at
CMU
Complemented by functional programming course in ML
See http://c0.typesafety.net

Concurrent C0: session-type message-passing concurrency
[Willsey et al.’16]

Examples from this tutorial
Many more examples, plus others in progress
svn co https://svn.concert.cs.cmu.edu/c0

User guest, pwd c0c0ffee

See c0/cc0-concur/README-concur.txt

Requires Standard ML (SML/NJ or mlton)

Compiles to C (or Go)

Other Ongoing Research

SILL: functional instantiation of ideas [Toninho et al.’13]
[Toninho’15] [Griffith & Pf’15]

Includes polymorphism and subtyping, not yet sharing

Adjoint logic [Reed’09]

Allows linear, affine, strict, and structural modes
Uniform concurrent semantics without sharing
[Chargin et al.’17]

Concurrent contracts [Gommerstadt et al.’18]

Concurrent type theory [Caires et al.’12]

A new foundation of object-oriented programming
[Balzer & Pf’15]

Automata and transducers in subsingleton fragment
[DeYoung & Pf’16]

Fault tolerance

Related Work (Small Sample)

Seminal work on session types
[Honda’93] [Honda, Vasconcelos & Kubo’98]

Subtyping [Gay & Hole’05]

Refinement types [Griffith & Gunter’13]

Classical linear logic and session types [Wadler’12]
[Toninho et al.’16]

Links language [Lindley et al.’06–]

Multiparty session types [Honda,Yoshida et al.’07–]

Scribble protocol language [Yoshida et al.’09–]

ABCD project [Gay, Wadler & Yoshida’13–’18]

Conclusion

From (linear) logical origins to a new foundation for
statically typed message-passing concurrency

Primitives are not quite those of the π-calculus

Simple, expressive, elegant, easy to use

Robust across paradigms

Functional (SILL, Links)
Imperative (Concurrent C0)
Object-oriented (Mungo)
Language agnostic (Scribble)

